Question: I am working on a physics reserach project for school and I have run into some troubles working Mathematica. I am a fairly inexperienced mathematica user so any help would very much be appreciated.

I need to find the roots of the transcendental equation 
$$\zeta\_n \tan(\zeta) - \sqrt{R^2-\zeta^2\_n}=0$$
and then collect them into a list, $\{\zeta\_n\}$, which I can sum over.

FindRoot works but only finds one root at a time. For example $R^2=18$ gives

> 
> FindRoot[Sqrt[18. - zeta^2] - zeta Tan[zeta] == 0, {zeta, 3}]
> 
> 
> {zeta -> 3.66808}
>  From a plot of the function I know there should be two roots for this particular value of $R$, however FindRoot only gives one.
> 
> 
> 

I have had no luck with NSolve or Solve either. 

> 
> NSolve[xi[zeta, Erg4]- zeta\*Tan[zeta] == 0, zeta]
> 
> 
> NSolve[Sqrt[18. - zeta^2] - zeta Tan[zeta] == 0, zeta]
> 
> 
> 

Also once I have succeeded in finding all roots how does one put them in an indexed set ${\zeta\_n}$.

Answer: 
```
With[{maxR = 10},
 Manipulate[
  expr = Sqrt[R^2 - zeta^2] - zeta Tan[zeta];
  zero = zeta /. NSolve[{expr == 0,
       R^2 - zeta^2 >= 0}, zeta] // Quiet;
  Column[{
    Plot[expr, {zeta, -maxR, maxR},
     Exclusions -> {Cos[zeta] == 0},
     PlotRange -> {{-maxR, maxR}, {-25, 40}},
     Epilog -> {Red, AbsolutePointSize[4],
       Point[{#, 0} & /@ zero]},
     AspectRatio -> 1],
    zero},
   Alignment -> Center],
  {{R, Sqrt[18]}, 0, maxR, Appearance -> "Labeled"}]]

```

![enter image description here](https://i.stack.imgur.com/8xcBF.png)
After a slight rearrangement, your equation is a good candidate for the Chebyshev approach, as detailed [here](https://mathematica.stackexchange.com/a/89692) and [here](https://mathematica.stackexchange.com/a/92042):

```
r = Sqrt[18]; f = Sqrt[r^2 - #^2] Cos[#] - # Sin[#] &;
n = 32;
cnodes = Rescale[N[Cos[Pi Range[0, n]/n], 20], {-1, 1}, {-r, r}];
cc = Sqrt[2/n] FourierDCT[f /@ cnodes, 1];
cc[[{1, -1}]] /= 2;

colleague = SparseArray[{{i_, j_} /; i + 1 == j :> 1/2,
                         {i_, j_} /; i == j + 1 :> 1/(2 - Boole[j == 1])},
                        {n, n}] -
            SparseArray[{{i_, n} :> cc[[i]]/(2 cc[[n + 1]])}, {n, n}];

rts = Sort[Select[DeleteCases[
           Rescale[Eigenvalues[colleague], {-1, 1}, {-r, r}],
                   _Complex | _DirectedInfinity], Abs[#] <= r &]];

Plot[x Tan[x] - Sqrt[r^2 - x^2], {x, -r, r},
     Epilog -> {Directive[Red, PointSize[Medium]], 
                Point[Transpose[PadRight[{rts}, {2, Automatic}]]]},
     Exclusions -> {Cos[x] == 0}]

```

![there's the roots!](https://i.stack.imgur.com/PDr9L.png)